



Introdução ao mundo do Desenvolvimento

Material das Aulas

Algoritmos

O que é?

Entrada → Processamento → Saída

Desafios e práticas

Hello World

Variáveis e Operações

O que são variáveis

Exercícios e praticando

Exemplo de Contas

Exemplo Idade em Dias

Decisões

O que é?

Exemplo com If e Else

Operador AND (&&)

Operador OR (|)

Exemplo com If e Else aninhado

Repetições

O que é?

Exemplo While

Exemplo com For

URI 1073 V1

URI 1073 V2

Material das Aulas

- [Portugol WebStudio](#)
- [Slides](#)

Algoritmos

O que é?

- É uma sequência de passos para solucionar um problema
- Sequência de comandos para transformar uma informação inicial em uma informação final.
- A **ordem dos passos** é muito importante
- São roteiros de como podemos executar uma tarefa
- Exemplo: fazer bolo, montar uma cadeira etc.

Entrada → Processamento → Saída

- Entrada é tudo que precisamos informar ao algoritmo, é uma informação bruta
- Processamento é nosso roteiro, sequência de passos.
- Saída é uma outra informação gerado a partir do conjunto da entrada e processamento.
- Exemplo no caso de fazer bolo: Ingredientes, Modo de Preparo e Bolo
- O formalismo é fundamental também para o algoritmo.



A ordem das instruções é fundamental para ter o resultado esperado, pois **a máquina só saber executar as coisas de forma sequencial.**

Desafios e práticas

- Vamos resolver problemas no [Portugol Webstudio](#)

Hello World

- Instruções uma após a outra.

```

1 programa {
2     funcao inicio() {
3         escreva("Oi, meu primeiro programa!\n")
4         escreva("Agora vai!\n\n")
5         escreva("Virei Hacker!! Partiu invadir a NASA!!")
6     }
7 }
8

```

```
Oi, meu primeiro programa!  
Agora vai!  
  
Virei Hacker!! Partiu invadir a NASA!!  
Programa finalizado.
```

Variáveis e Operações

O que são variáveis

- São áreas de memórias onde armazenamos algo, como se fossem caixinhas.
- O conteúdo dele pode variar.
- Podemos qualificar a informação atribuindo um tipo de dado
- Podemos também quantificar dizendo quanto espaço de memória ocupa nosso conteúdo

Em Portugal

- O conteúdo pode ser Inteiro, Real e Cadeias
- Sem aspas é o conteúdo da variável

```
funcao inicio() {  
    inteiro var1  
    real    var2  
    cadeia  var3
```

```
escreva("Por favor, digite um numero inteiro:")  
leia(var1)  
escreva("Agora digite um numero real:")  
leia(var2)  
escreva("Finalmente digite um texto:")  
leia(var3)  
escreva("Voce digitou os seguintes valores\n")  
escreva(var1 + "\n" + var2 + "\n" + var3)
```

Exercícios e praticando

- Podemos praticar usando sites como:
 - Beecrowder

Exemplo de Contas

- Número inteiro, não pega números decimais.

```

programa {
    funcao inicio() {
        inteiro valor1, valor2, res
        escreva("digite um valor:")
        leia(valor1)
        escreva("digite outro valor:")
        leia(valor2)

        res = valor1 + valor2
        escreva("Resultado da soma = "+res+"\n")

        res = valor1 - valor2
        escreva("Resultado da subtracao = "+res+"\n")

        res = valor1 * valor2
        escreva("Resultado da multiplicacao = "+res+"\n")

        res = valor1 / valor2
        escreva("Resultado da divisao = "+res+"\n")
    }
}

```

```

digite um valor:20
digite outro valor:3
Resultado da soma = 23
Resultado da subtracao = 17
Resultado da multiplicacao = 60
Resultado da divisao = 6

```

Exemplo Idade em Dias

- Trabalhamos sempre com o resto da divisão para pegar os valores de meses e dias.

```

programa {
    funcao inicio() {
        inteiro idade, qtAnos, qtMeses, qtDias, sobra

        escreva("Digite a idade em dias:")
        leia(idade)

        qtAnos = idade / 365

        sobra = idade % 365

        qtMeses = sobra / 30

        qtDias = sobra % 30

        escreva(qtAnos + " ano(s)\n")
        escreva(qtMeses + " mes(es)\n")
        escreva(qtDias + " dia(s)\n")
    }
}

```

Decisões

O que é?

- São caminhos de desvio do fluxo sequencial do nosso código
- Testes lógicos que seguem Verdadeiro ou Falso, e a partir dessa comparação seguem um fluxo de código.
- A partir de comparações relacionais seguimos para um caminho retornando verdadeiro ou falso.

Exemplo com If e Else

- Caso verdadeiro vai seguir para um código, caso falso segue para outro código

```
programa {  
    funcao inicio() {  
        real nota1, nota2, media  
  
        escreva("Digite a nota 1:")  
        leia(nota1)  
        escreva("Digite a nota 2:")  
        leia(nota2)  
  
        media = (nota1 + nota2) / 2  
  
        escreva("A media vale = "+media+"\n")  
  
        se ( media >= 6.0 ){  
            escreva("Parabens voce foi aprovado(a)!\n")  
        }  
        senao{  
            escreva("Obaa nos vemos ano que vem!\n")  
        }  
  
        escreva("Fim do programa\n")  
    }  
}
```

Operador AND (&&)

- Os dois lados da comparação precisam ser verdadeiro para a expressão ser verdadeira

Operador OR (|)

- Basta apenas um lado da comparação ser verdadeiro para a expressão ser verdadeira.

Exemplo com If e Else aninhado

- A cada negativa do nosso código, vamos para outra parte dele, até a condicional ser cumprida.
- A partir da condicional temos ações sendo executadas
- Poderíamos fazer com `Switch Else` também

```

se (media >= 9 e media <= 10){ // vou testar se é conceito A
    escreva("conceito A")
}
senao{ // ainda pode ser B, C, D ou R
    se (media >= 8 e media < 9){ // teste se é B
        escreva("conceito B")
    }
    senao{ // ainda pode ser C, D ou R
        se (media >= 6 e media < 8){ // teste se é C
            escreva("conceito C")
        }
        senao { // ainda pode ser D ou R
            se (media >= 6 e media < 6){ // teste se é D
                escreva("conceito D")
            }
            senao{ // so pode ser R
                escreva("Reprovado")
            }
        }
    }
}

```

Repetições

O que é?

- Fazemos algoritmos para lidar com tarefas repetitivas.
- Todas as instruções são executadas na sequência que são escritas
- Nas repetições conseguimos alterar o fluxo dos códigos.
 - `Do While`
 - `For..`
- Enquanto determinada condição for satisfeita o código continua sendo executado.

Exemplo While

- Vai executar enquanto a variável de controlar (contador) for menor que 10.
- Fará uma tabuada.

```

programa {
    funcao inicio() {
        inteiro num, contador, resultado

        escreva("qual tabuada? ")
        leia(num)

        contador = 0

        enquanto (contador <= 10){
            resultado = num * contador
            escreva(num + " x " + contador + " = " + resultado + "\n")
            contador = contador + 1
        }
    }
}

```

Exemplo com For

- Temos uma variável de controle e o incremento já na condicional.
- Executará o passo a passo que estiver dentro da função, enquanto cumprir a condicional, ou seja enquanto a expressão for verdadeira.

```

programa {
    funcao inicio() {
        inteiro num, contador, resultado

        escreva("qual tabuada usando o PARA? ")
        leia(num)

        para (contador = 0 ; contador <= 10; contador = contador + 1){
            resultado = contador * num
            escreva(num + " x " + contador + " = " + resultado + "\n")
        }
    }
}

```

URI 1073 V1

- Dessa vez usamos com estruturas de decisão e repetição

```

programa {
    funcao inicio() {
        inteiro n, numero, quadrado

        escreva("digite o numero final ")
        leia(n)

        para (numero = 1 ; numero <= n; numero = numero + 1){
            se (numero % 2 == 0){
                quadrado = numero * numero
                escreva(numero + "^2 = " + quadrado + "\n")
            }
        }
    }
}

```

URI 1073 V2

- Dessa vez usamos o For, de decisão.

```

programa {
    funcao inicio() {
        inteiro n, numero

        escreva("Qual o numero final? ")
        leia(n)

        para( numero = 2 ; numero <= n ; numero = numero + 2){
            //quadrado = numero * numero
            escreva(numero + "^2 = " + (numero * numero) + "\n")
        }
    }
}

```



Desenvolva sua própria lógica para resolução de problemas.